DOCUMENT RESUME

ED 049 306 TM 000 506

AUTHOR	Jensema, Carl
TITLE	A Review of a Rotation to Obtain Maximum Similarity and Simple Structure Among Factor Patterns.
INSTITUTION	Washington Univ., Seattle. Bureau of Testing.
REPORT NO	BTP-176
PUB DATE	[71]
NOTE	19p.
EDRS PRICE	EDRS Price MF-$0.65 HC-$3.29
DESCRIPTORS	Algorithms, *Computer Programs, Correlation, *Factor Analysis, *Factor Structure, Oblique Rotation, Orthogonal Rotation, Research Methodology, *Statistical Analysis, *Transformations (Mathematics)
IDENTIFIERS	*Bloxoms Rotation Technique

ABSTRACT

Under some circumstances, it is desirable to compare the factor patterns obtained from different factor analyses. To date, the best method of simultaneously achieving simple structure and maximum similarity is the technique devised by Bloxom (1968). This technique simultaneously rotates different factor patterns to maximum similarity and varimax simple structure and is applicable when the number of factors and variables are the same for each subpopulation. The technique is described and the computer algorithms are given. A numerical example compares the results of different factor analyses. If the limitations of the method are observed, Bloxom's rotation technique has the potential for wide application in studies involving comparison of factor patterns. A small loss in simple structure through the use of Bloxom's method often pays off in factors which can be much more meaningfully compared. (CK)

ED 049 306

A review of a rotation to obtain maximum similarity and simple structure among factor patterns[1]

Carl Jensema

University of Washington

Under some circumstances it is desirable to compare the factor patterns obtained from different factor analyses. When the number of factors and variables are the same for each subpopulation, it is often possible to apply a rotation technique developed by Bloxom which simultaneously rotates different factor patterns to maximum similarity and varimax simple structure. The technique is described, computer algorithms are outlined, and a numerical example is given.

Bureau of Testing Project: 176

[1]The author wishes to express his deep appreciation to Dr. Vern Urry and Dr. Clifford E. Lunneborg. Their advice and helpful insights have been invaluable.

TM 000 506

A review of a rotation to obtain maximum similarity and simple structure among factor patterns

Carl Jensema

University of Washington

Introduction

Factorial invariance is important in a variety of common research situations involving factor analysis. For example, an investigator may give the same questionnaire or test to several subpopulations and wish to know how stable the underlying factors are across these subpopulations. Alternatively, a researcher may give several similar questionnaires or tests to the same group and wish to determine whether they all measure the same factors.

In both of these two general situations or variations of them, the investigator has a choice: He may rotate each of his factor pattern matrices to simple structure or he may rotate them to maximum similarity with each other. A multitude of rotational techniques, both orthogonal and oblique, have been proposed for achieving simple structure transforms which are unique for a given data set. Methods for achieving maximum similarity between factor patterns are somewhat less numerous. They have developed along two general lines. The first is canonical correlations, which rotates correlational matrices to mutual maximum similarity. The second approach involves rotation to maximum similarity in relation to a target matrix. This later method is commonly known as Procrustes rotation.

As mentioned above, rotation techniques which are designed to achieve simple structure for a given data set tend to produce highly unique results. Comparison of different factor structure is usually difficult unless all factors are extremely similar. On the other hand, if rotation is done to achieve maximum similarity between factor patterns, simple structure is almost always destroyed and meaningful interpretation is nearly impossible. Under certain conditions Procrustes rotation may produce results which are an exception to this general observations, but even here, the target matrix is arbitrary and there is no way of ascertaining if it is the best possible one.

It is highly desirable to have a rotational technique which will achieve both maximum simple structure and maximum similarity between different factor patterns. This has been recognized and several psychometricians have considered the problem. An article by Ahmavaraa (1954) provided some basic groundwork on the topic. Meredith (1964a) generalized Ahmavaraa's work using Lawley's Selection theorem and proposed a rotational technique (Meredith, 1964b). Unfortunately, Meredith's solution has some rather severe limitations. It requires that the regression of the variates being factored on the variates defining the subpopulations be linear and homoscedastic. The solution collapses if these conditions are not met.

Cliff (1966) considered the problems of 1) rotating two solutions to maximum similarity and 2) rotating a solution orthogonally to a specified target matrix. He suggested a sequential procedure to obtain maximally similar simple structure solutions for two factor matrices.

The problem with this treatment is that it may tend to be cumbersome and time consuming.

To date, the best method of simultaneously achieving simple structure and maximum similarity is the technique of Bloxom (1968). He reversed Meredith's approach and instead of assuming a constant factor pattern matrix while allowing the factor axore variance-covariance matrices to vary, he allowed the factor pattern matrices to vary across subpopulations while holding the factor score variance-co-variance matrix arbitrarily constant. This effectively circumverted the limitations in Meredith's procedure while being more convenient and generally applicable than the treatment suggested by Cliff. However, Bloxom's method appears to have escaped the general notice of psychologists. This paper will attempt to correct that situation by outlining the method and demonstrating its usefulness.

General Method

Rotation of factor patterns is usually described as:

(1) $$b = aH$$

where a is the factor loading matrix, H is the orthonormal transformation matrix, and b is the rotated simple structure matrix. If there are m subpopulation factor pattern matrices, each having the same number of variables and factors, the problem is to determine a transformation matrix, H, for each of the m pattern matrices such that the resulting rotated matrices will be maximally similar and yet conform to the requirements of simple structure.

The Varimax method of rotation, as described by Horst (1965), maximizes

(2) $$\emptyset = \operatorname{tr}\left[b^{(2)'} \left(I - \frac{11'}{n}\right) b^{(2)} \right]$$

where $b^{(2)}$ is a matrix of the squared elements of the rotated loadings and $\left(I - \frac{11'}{n}\right)$ is an indempotent matrix. Bloxom has generalized this criterion across m loading matrices. His index of similarity between two factor loading matrices, ${}_ib$ and ${}_jb$, is given by

(3) $${}_{ij}\emptyset = \operatorname{tr}\left[{}_ib^{(2)'} \left(I - \frac{11'}{n}\right) {}_jb^{(2)} \right] ,$$

which gives the sum of covariances of corresponding columns of ${}_ib^{(2)}$ and ${}_jb^{(2)}$.

By combining (2) and (3), Bloxom obtained

(4) $$\psi = \sum_{i=1}^{m} \sum_{j=1}^{m} \left[{}_ib^{(2)'} \left(I - \frac{11'}{n}\right) {}_jb^{(2)} \right]$$

as an expression of similarity between all m factor loading matrices and the extent to which the varimax criterion is maximized for each of the m loading matrices. By maximizing (4) with the constraint that

(5) $${}_iH'\,{}_iH = {}_iH\,{}_iH' = I,$$

maximum similarity and maximum simple structure are achieved among the m rotated factor loading matrices.

Computation

Initially, each of the k x k transformation matrices are set to identity matrices. The n x n indempotent matrix, $\left(I - \frac{11'}{n}\right)$, is calculated by setting each diagonal element to $\frac{n-1}{n}$ and each off diagonal element to $\frac{-1}{n}$ where n is the number of variables. After these initializations are completed the iteration procedure is begun. Taking the unrotated loadings, $_ia$, of each of the m matrices, the rotated matrices, $_ib$, are obtained from

(6) $$_ib = {}_ia \; {}_iH$$

Squaring each element in each $_ib$ matrix, we calculate

(7) $$\psi = \sum_{i=1}^{m} \sum_{j=1}^{m} \operatorname{tr} \left[{}_ib^{(2)'} \, V \, {}_jb^{(2)} \right]$$

where V is the indempotent matrix, $\left(I - \frac{11'}{n}\right)$. This is the criterion we seek to maximize. Iterations will be halted when ψ stabilizes.

Continuing the iterative procedure, equations (8) through (11) are looped through m times, once for each transformation matrix. Calculating a $_i\beta$ matrix as

(8) $$_i\beta_{.k} = \left[\sum_{j\neq i}^{m} D_{ib.k} \, V \, {}_jb^{(2)}_{.k} \right] + \left[2 D_{ib.k} \, V \, {}_ib^{(2)}_{.k} \right]$$

multiplication by the unrotated loading matrix, $_ia$, gives

(9) $$_ia \, _i\beta \, .$$

All roots and vectors of the minor product moment of this matrix are obtained, as indicated by

(10) $$_iq \, _id^2 \, _iq' = \left({}_i\beta' \, {}_ia \right) \left({}_ka' \, {}_i\beta \right)$$

From this the new estimate of the transformation matrix is calculated as

$$(11 \qquad {}_iH = \left({}_ia' \; {}_i\ell\right)\left({}_iq \; {}_id^{-1} \; {}_iq'\right) \quad .$$

As previously indicated, equations (8) through (11) are carried out for each of m loading matrices. When this has been accomplished, equation (6) is returned to and a new iteration is begun. This process is continued until successive iterations indicate that ψ has stabilized and final rotated matrices ${}_ib$ are the best possible.

Once the maximally similar varimax patterns have been found, it is usually desirable to have some measure of the degree of factorial similarity that has been achieved. The most common statistical index of similarity is the cosine between two factor vectors. The relationship between two factor pattern matrices, A and B, is given by

$$(12) \qquad AT = B + E$$

where T is a transformation matrix and E is a matrix of error terms. The T matrix may be viewed as an estimate of the cosines between factor pattern matrices A and B, and, as such, can be considered an index of factorial similarity. Schönemann (1966) has given a solution for T utilizing the constraints that

$$(13) \qquad T'T = TT' = I$$

and

$$(14) \qquad \mathrm{tr}\,(E'E) = \min.$$

Briefly, Schönemann's T solution is calculated in the following manner:

$$(15) \qquad S = A'B,$$

$$(16) \qquad S'S = VD_sV' ,$$

(17) $$S S' = W D_s W'$$

and

(18) $$T = W V'.$$

The W matrix of eigennectors in (17) must be reflected before (18) is computed. This may be done by obtaining the diagonal

(19) $$D_s^{\frac{1}{2}} = W'SV$$

and examining $D_s^{\frac{1}{2}}$ for negative elements. If an element of $D_s^{\frac{1}{2}}$ is negative, reflect the corresponding column in W. When computed as described above, the T matrix has been found to be a useful index of factorial similarity for evaluating the output of Bloxom's rotation.

Application

Semantic differential data were used to demonstrate the usefulness of Bloxom's method in comparing the results of different factor analyses. Each of 176 subjects responded to the same 25 semantic differential bipolar word pairs for each of 3 concepts. These concepts were "Blind person", "Deaf person", and "Amputee". A principle components factor analysis was performed on each concept and three factors per concept were extracted. The factors for each concept were rotated with varimax and then with Bloxom's rotation. Table 1 gives the factor patterns obtained with each of these methods. Visual examination of the Bloxom rotation patterns readily demonstrates the close similarity between the three factors involved in each concept. This similarity is not nearly so apparent in the varimax factor patterns.

Further evidence of similarity between factors is given by the cosines in Table 2. These provide a more precise measure of factorial similarity than simple visual examination. Cosines can take values ranging from +1.00 to -1.00. A cosine of +1.00 indicates perfect alignment of factor vectors, .00 represents orthogonal factors, and -1.00 means the poles of the factor vectors are reversed. From Table 2 it can be seen that the Bloxom method of rotation does an excellent job of producing factor patterns which can be compared across concepts.

As with any statistical technique, Bloxom's rotation has limitations. Bloxom has pointed out that in some cases the method gives preferential treatment for factors having a large number of high loadings. He suggested that this might be caused by a covariance "tug-of-war" between larger and smaller factors. If this happens, the factor patterns may tend to be varimax-covarimax for larger factors and varimax for smaller factors. Caution must be used in applying the method where factors within subpopulations are not of comparable magnitude.

If the limitations of the method are observed, Bloxom's rotation technique has the potential for wide application in studies involving comparison of factor patterns. A small loss in simple structure through the use of Bloxom's method often pays off in factors which can be much more meaningfully compared.

Table 1

Factor Patterns Obtained with Two Rotation Methods

	Varimax									Bloxom								
	Blind Person			Deaf Person			Amputee			Blind Person			Deaf Person			Amputee		
	1	2	3	1	2	3	1	2	3	1	2	3	1	2	3	1	2	3
1	18	54	00	02	47	05	23	00	62	23	50	16	15	45	02	28	60	00
2	52	-05	19	28	-18	54	05	71	10	21	-12	50	21	-22	55	08	09	71
3	13	16	50	66	-16	14	65	20	30	53	-06	07	58	-32	16	68	25	18
4	30	04	50	58	01	20	72	20	18	50	-17	25	56	-14	21	74	12	18
5	68	12	03	22	13	55	02	58	-02	15	09	67	24	10	55	04	-02	58
6	22	-15	-04	07	-48	16	04	18	-80	-07	-12	24	-07	-47	18	-02	-80	18
7	09	-16	65	55	-04	34	79	03	-10	54	-41	03	51	-17	35	78	-17	00
8	66	06	20	22	-10	63	11	73	-18	28	-03	63	18	-12	64	12	-19	73
9	-16	30	31	05	71	10	24	-16	-64	38	15	-20	25	67	07	28	62	-16
10	48	-43	49	37	-41	50	16	70	-28	32	-59	44	24	-47	53	15	-30	69
11	43	20	54	67	16	28	68	33	24	62	-04	36	68	-01	29	71	16	31
12	15	-56	-02	06	53	09	43	-29	26	-22	-51	17	21	50	07	44	23	-30
13	57	-38	18	39	-33	38	-06	75	-01	08	-42	56	28	-40	41	-04	01	75
14	45	-08	41	16	-13	63	13	75	-16	39	-24	41	11	-14	64	14	-18	74
15	54	-18	17	06	02	69	00	79	04	14	-24	53	06	04	69	03	03	79
16	46	46	20	64	21	16	68	06	26	41	34	41	67	03	17	70	20	04
17	21	-55	-09	13	36	-16	-26	37	-41	-28	-47	24	23	30	-17	-28	-39	37
18	59	28	-06	09	18	66	06	70	-13	12	27	58	12	18	65	08	-14	70
19	69	24	-13	11	02	79	-11	72	05	05	27	69	10	04	79	-08	06	72
20	28	64	13	35	64	10	46	04	58	46	50	23	51	52	08	50	54	02
21	-17	12	50	44	33	01	58	-17	11	48	-09	-23	51	20	01	58	07	-18
22	19	37	52	58	37	07	79	-17	-07	64	13	12	66	20	07	78	-14	-19
23	27	46	49	69	33	14	78	16	11	66	22	20	75	13	14	79	04	13
24	52	04	22	23	12	64	20	63	-10	28	-05	50	24	03	64	21	-12	62
25	48	-04	38	67	02	23	56	32	30	38	-19	44	64	-19	25	60	25	30

Note--Loadings with absolute values greater than .40 are underlined

Table 2

Cosines Between Factors

Varimax

	Blind Person			Deaf Person			Amputee			Blind Person			Deaf Person			Amputee		
	1	2	3	1	2	3	1	2	3	1	2	3	1	2	3	1	2	3
Blind Person 1	100	00	00	14	-02	99	04	99	10	100	00	00	100	06	-01	99	13	01
2		100	00	08	100	01	20	-11	97		100	00	-06	100	00	-13	100	-08
3			100	99	-08	-14	98	-02	-20			100	01	00	100	-02	08	100
Deaf Person 1				100	00	00	98	10	-15				100	00	00	100	03	01
2					100	00	17	-17	97					100	00	-03	99	-13
3						100	-07	98	18						100	-01	13	99
Amputee 1							100	00	00							100	00	00
2								100	00								100	00
3									100									100

Bibliography

Bloxom, Bruce. Factorial rotation to simple structure and maximum similarity. Psychometrika, 1968, 33, 237-247.

Cliff, Norman. Orthogonal rotation to congruence. Psychometrika 1966, 31, 33-42.

Horst, Paul. Factor analysis of data matrices. San Francisco: Holt, Rinehart, and Winston, 1965.

Meredith, William. Notes on factorial invariance. Psychometrika 1964, 29, 177-186 (a).

Meredith, William. Rotation to achieve factorial invariance. Psychometrika 1964, 29, 187-206 (b).

Schönemann, Peter H. A generalized solution of the orthogonal procrustes problem. Psychometrika 1966, 31, 1-10.

"MAXSIM"

Program Setup Instructions

A. Title Card--Any alphanumeric message to be printed at beginning of output.

B. Problem Card

Column	Item
1 - 7	Punch "PROBLEM"
8 - 9	Number of matrices (Max = 10)
10 - 11	Number of variables in each Matrix (Max = 50)
12 - 13	Number of factors in each Matrix (Max = 15)
14 - 16	Maximum number of iterations to be allowed.
17 - 21	Iteration tolerance (try 00100 or 00010)
22 - 23	Number of input unit

C. Input Format Card. Any F-type format

D. Data--Matrices should be read in one at a time with factors arranged in columns and variables in rows

This is a repeating program. Steps A through D can be repeated as often as desired.

E. At the end of all problems put a card with "FINISH" punched in columns 1-6.

```
RUN(S).
SETCORE.
LGO.
-00
      PROGRAM MAXSIM(INPUT,OUTPUT,PUNCH,TAPE5=INPUT,TAPE6=OUTPUT,TAPE7=
     *PUNCH)
      DIMENSION A(10,50,15),B(10,50,15),BV(10,15,50),H(10,15,15),
     *V(50,50),BB(50,15),AB(15,15),BAAB(30,15),E(15),TITLE(8),FMT(8)
C     THIS PROGRAM WAS WRITTEN BY CARL JENSEMA IN JUNE,1970.
C     IT TRANSFORMS FACTOR LOADING MATRICES OF DIFFERENT GROUPS TO
C     MAXIMALLY SIMILAR ORTHOGONAL SIMPLE STRUCTURES.
C     THE METHOD IS OUTLINED BY BLOXOM IN PSYCHOMETRIKA
C     (VOL.33,NO.2,JUNE,1968). THE SIMPLE STRUCTURE CRITERION
C     IS THE SAME AS FOR VARIMAX TRANSFORMATIONS.
C     M=NUMBER OF GROUPS (MAX=10)
C     NV=NUMBER OF VARIABLES (MAX=50)
C     NF=NUMBER OF FACTORS (MAX=15)
C     A=ORIGINAL FACTOR LOADING MATRICES
C     H=TRANSFORMATION MATRICES
C     B=NEW FACTOR LOADING MATRICES
10    READ(5,1) TITLE,PROBLEM,M,NV,NF,MAX,TOL,NT,FMT
1     FORMAT(8A10/A7,3I2,I3,F5.5,I2/8A10)
      IF(TITLE(1).EQ.10HFINISH    ) GO TO 10000
      IF(FMT(1).EQ.10HFINISH    ) GO TO 10000
      IF(PROBLEM.EQ.7HFINISH ) GO TO 10000
      IF(PROBLEM.NE.7HPROBLEM) GO TO 10
      WRITE(6,2) TITLE,M,NV,NF,MAX,TOL,NT,FMT
2     FORMAT(45H1UNIVERSITY OF WASHINGTON BUREAU OF TESTING
     1//76H PROGRAM TO TRANSFORM FACTOR LOADINGS TO MAXIMALLY SIMILAR SI
     *MPLE STRUCTURES    //1X,8A10
     2//19H NUMBER OF GROUPS = 17X,I3
     3//36H NUMBER OF VARIABLES IN EACH GROUP = I3
     4//36H NUMBER OF FACTORS IN EACH GROUP =   I3
     5//36H MAXIMUM NUMBER OF ITERATIONS =      I3
     6//22H ITERATION TOLERANCE = 14X,F9.5
     7//13H INPUT UNIT = 23X,I3
     8//15H INPUT FORMAT = 10X,8A10)
C     READ IN ORIGINAL LOADINGS.
      DO 20 I=1,M
      WRITE(6,3) I
3     FORMAT(5(/),28H ORIGINAL LOADINGS FOR GROUP I3)
      DO 20 J=1,NV
      READ(NT,FMT)(A(I,J,K),K=1,NF)
      WRITE(6,4) J,(A(I,J,K),K=1,NF)
4     FORMAT(//9H VARIABLEI3,(/ 15F9.3))
20    CONTINUE
      WRITE(6,5)
5     FORMAT(5(/),32H COSINES BETWEEN INPUT MATRICES.     )
      CALL COSINES(A,BAAB,AB,E,TOL,M,NV,NF)
      WRITE(6,6)
6     FORMAT(5(/),38H SUCCESSIVE ITERATION CRITERION VALUES   )
C     CALCULATE IDEMPOTENT MATRIX USED IN CRITERION FUNCTION.
C     THE CRITERION IS THE SAME AS THE ONE USED IN VARIMAX TRANSFORMS.
      DO 40 I=1,NV
      DO 30 J=1,NV
      V(I,J)=0.0-1.0/NV
```

```
30      CONTINUE
        V(I,I)=1.0-1.0/NV
40      CONTINUE
C       SET TRANSFORMATION MATRIX H TO AN IDENTITY MATRIX.
        DO 60 I=1,M
        DO 60 J=1,NF
        DO 50 K=1,NF
        H(I,J,K)=0.0
50      CONTINUE
        H(I,J,J)=1.0
60      CONTINUE
        LOOP=-1
        PSI=0.0
C       BEGIN ITERATION LOOP.
C       MULTIPLY ORIGINAL LOADINGS BY TRANSFORMATION MATRIX.
C       THIS IS EQUATION 1 OF BLOXOMS ARTICLE.
70      LOOP=LOOP+1
        DO 80 I=1,M
        DO 80 J=1,NV
        DO 80 K=1,NF
        B(I,J,K)=0.0
        DO 80 L=1,NF
        B(I,J,K)=B(I,J,K)+A(I,J,L)*H(I,L,K)
80      CONTINUE
C       BEGIN CALCULATION OF CRITERION FUNCTION PSI.
C       THIS IS GIVEN IN EQUATION 5 OF BLOXOMS ARTICLE.
        DO 90 I=1,M
        DO 90 J=1,NF
        DO 90 K=1,NV
        BV(I,J,K)=0.0
        DO 90 L=1,NV
        BV(I,J,K)=BV(I,J,K)+(B(I,L,J)**2)*V(L,K)
90      CONTINUE
        OLDPSI=PSI
        PSI=0.0
        DO 100 I=1,M
        DO 100 J=1,M
        DO 100 K=1,NF
        DO 100 L=1,NV
        PSI=PSI+BV(I,K,L)*(B(J,L,K)**2)
100     CONTINUE
C       PSI IS NOW THE CRITERION FUNCTION VALUE SPECIFIED BY EQUATION 5.
        WRITE(6,7) PSI
7       FORMAT(F15.6)
C       CHECK TO SEE IF ITERATION TOLERANCE HAS BEEN REACHED.
        IF(ABS(PSI-OLDPSI).LE.TOL) GO TO 200
C       CHECK TO SEE IF THE MAXIMUM NUMBER OF ITERATIONS HAS BEEN REACHED.
        IF(LOOP.GE.MAX) GO TO 200
C       BEGIN INNER LOOP TO GET A NEW TRANSFORMATION MATRIX FOR EACH GROUP
        DO 190 II=1,M
        DO 110 I=1,NF
        DO 110 J=1,NV
        BB(I,J)=0.0
110     CONTINUE
        DO 120 I=1,M
        DOUBLE=1.0
```

```
      WRITE(6,11)
11    FORMAT(5(/),33H COSINES BETWEEN OUTPUT MATRICES.     )
      CALL COSINES(B,BAAB,AB,E,TOL,M,NV,NF)
      GO TO 10
10000 WRITE(6,12)
12    FORMAT(20(/),12H END OF JOB.)
      STOP
      END

      SUBROUTINECOSINES(B,BAAB,AB,E,TOL,M,NV,NF)
C     THIS SUBROUTINE CALCULATES ORTHOGONAL PROCRUSTES TRANSFORMATION
C     (COSINE) MATRICES.  THE METHOD COMES FROM SCHONEMANN (PSYCHOMETRIKA,
C     MARCH, 1966).
      DIMENSION B(10,50,15),BAAB(30,15),ABBA(30,15),AB(15,15),E(15)
      DO 70 II=1,M
      DO 70 JJ=II,M
C     OBTAIN MINOR PRODUCT OF TWO LOADING MATRICES.
      DO 10 I=1,NF
      DO 10 J=1,NF
      AB(I,J)=0.0
      DO 10 K=1,NV
      AB(I,J)=AB(I,J)+B(II,K,I)*B(JJ,K,J)
10    CONTINUE
C     OBTAIN MINOR AND MAJOR PRODUCTS OF AB.
      DO 20 I=1,NF
      DO 20 J=1,NF
      BAAB(I,J)=0.0
      ABBA(I,J)=0.0
      DO 20 K=1,NF
      BAAB(I,J)=BAAB(I,J)+AB(K,I)*AB(K,J)
      ABBA(I,J)=ABBA(I,J)+AB(I,K)*AB(J,K)
20    CONTINUE
C     OBTAIN ROOTS AND VECTORS.
      CALL JACSIM(BAAB,E,TOL,NF)
      CALL JACSIM(ABBA,E,TOL,NF)
C     BEGIN REFLECTION OF ABBA VECTORS.
C     (SEE P.8, BOTTOM PARAGRAPH OF SCHONEMANN ARTICLE.)
C     MULTIPLY TO OBTAIN WSV.
      DO 30 I=1,NF
      DO 30 J=1,NF
      ABBA(I,J)=0.0
      DO 30 K=1,NF
      KNF=K+NF
      ABBA(I,J)=ABBA(I,J)+ABBA(KNF,I)*AB(K,J)
30    CONTINUE
      DO 40 I=1,NF
      DO 40 J=1,NF
      AB(I,J)=0.0
      DO 40 K=1,NF
      KNF=K+NF
      AB(I,J)=AB(I,J)+ABBA(I,K)*BAAB(KNF,J)
40    CONTINUE
C     CHECK DIAGONAL OF WSV, REFLECT COLUMN IF DIAGONAL ELEMENT IS NEGATIVE.
      DO 50 I=1,NF
      IF(AB(I,I).GE.0.) GO TO 50
      DO 50 J=1,NF
```

```
      IF(I.EQ.II) DOUBLE=2.0
      DO 120 J=1,NF
      DO 120 K=1,NV
      TEMP=B(II,K,J)
      DO 120 L=1,NV
      BB(K,J)=BB(K,J)+TEMP*V(K,L)*(B(I,L,J)**2)*DOUBLE
120   CONTINUE
C     WE NOW HAVE THE BETA MATRIX GIVEN IN EQUATION 18.
C     MULTIPLY THE BETA MATRIX BY THE ORIGINAL LOADINGS,
C     THEN CALCULATE THE MINOR PRODUCT MOMENT OF THIS NEW MATRIX.
      DO 140 I=1,NF
      DO 140 J=1,NF
      AB(I,J)=0.0
      DO 140 K=1,NV
      AB(I,J)=AB(I,J)+A(II,K,I)*BB(K,J)
140   CONTINUE
      DO 150 I=1,NF
      DO 150 J=1,NF
      BAAB(I,J)=0.0
      DO 150 K=1,NF
      BAAB(I,J)=BAAB(I,J)+AB(K,I)*AB(K,J)
150   CONTINUE
C     DO A COMPLETE FACTORIZATION OF MATRIX BAAB.
      CALL JACSIM(BAAB,E,TOL,NF)
C     EIGENVECTORS ARE COLUMNS OF BAAB.
C     EIGENVALUES ARE IN E.
      DO 160 I=1,NF
      E(I)=1.0/SQRT(E(I))
160   CONTINUE
      DO 170 I=1,NF
      INF=I+NF
      DO 170 J=1,NF
      JNF=J+NF
      BAAB(I,J)=0.0
      DO 170 K=1,NF
      BAAB(I,J)=BAAB(I,J)+BAAB(INF,K)*E(K)*BAAB(JNF,K)
170   CONTINUE
      DO 180 I=1,NF
      DO 180 J=1,NF
      H(II,I,J)=0.0
      DO 180 K=1,NF
      H(II,I,J)=H(II,I,J)+AB(I,K)*BAAB(K,J)
180   CONTINUE
C     WE NOW HAVE NEW ESTIMATIONS OF THE TRANSFORMATION MATRICES.
C     END OF INNER LOOP
190   CONTINUE
C     END OF ITERATION LOOP
      GO TO 70
200   WRITE(6,8) LOOP
8     FORMAT(//,30H NUMBER OF ITERATIONS NEEDED = I5)
      DO 210 I=1,M
      WRITE(6,9) I
9     FORMAT(5(/),31H TRANSFORMED LOADINGS FOR GROUP I3)
      DO 210 J=1,NV
      WRITE(6,4) J, (B(I,J,K),K=1,NF)
210   CONTINUE
```

```
      JNF=J+NF
      ABBA(JNF,I)=ABBA(JNF,I)*(-1.)
50    CONTINUE
C     END OF REFLECTION.
C     OBTAIN TRANSFORMATION MATRIX BY MULTIPLYING VECTORS.
      DO 60 I=1,NF
      INF=I+NF
      DO 60 J=1,NF
      JNF=J+NF
      AB(I,J)=0.0
      DO 60 K=1,NF
      AB(I,J)=AB(I,J)+ABBA(INF,K)*BAAB(JNF,K)
60    CONTINUE
      WRITE(6,1) II,JJ
1     FORMAT(///22H COSINES BETWEEN GROUPI3,17H (ROWS) AND GROUPI3,
     *11H (COLUMNS). )
      DO 70 I=1,NF
      WRITE(6,2) I,(AB(I,J),J=1,NF)
2     FORMAT(//7H FACTOR,I3(/15F9.3))
70    CONTINUE
      RETURN
      END

      SUBROUTINE JACSIM(R,D,P,N)
C     THIS SUBROUTINE CALCULATES ALL ROOTS AND VECTORS OF A MATRIX
C     USING A JACOBI METHOD.
      DIMENSION R(30,15),D(15)
      N1=N+1
      N11=N-1
      N2=N*2
      DO 10 I=N1,N2
      DO 10 J=1,N
      R(I,J)=0.0
10    CONTINUE
      DO 20 I=1,N
      NI=N+I
      R(NI,I)=1.0
20    CONTINUE
      DO 90 L=1,100
      DO 30 I=1,N
      D(I)=R(I,I)
30    CONTINUE
      DO 50 I=1,N11
      I1=I+1
      DO 50 J=I1,N
      DR=R(I,I)-R(J,J)
      A=SQRT(DR**2+4.*R(I,J)**2)
      A=SQRT((A+DR)/(2.*A))
      B=SQRT(1.-A**2)
      C=SIGN(1.,R(I,J))
      DO 40 K=1,N2
      U=R(K,I)*A*C+R(K,J)*B
      R(K,J)=-R(K,I)*B*C+R(K,J)*A
      R(K,I)=U
40    CONTINUE
      DO 50 K=1,N
```

```
      U=R(I,K)*A*C+R(J,K)*B
      R(J,K)=-R(I,K)*B*C+R(J,K)*A
      R(I,K)=U
 50   CONTINUE
      DO 60 I=1,N
      D(I)=ABS(D(I)-R(I,I))
 60   CONTINUE
      S=0.0
      DO 70 I=1,N
      S=MAX1(S,D(I))
 70   CONTINUE
      DO 80 I=1,N
      D(I)=R(I,I)
 80   CONTINUE
      IF(S-P) 100,100,90
 90   CONTINUE
 100  RETURN
      END
```